

NAG Library Routine Document

F08ZSF (ZGGQRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08ZSF (ZGGQRF) computes a generalized QR factorization of a complex matrix pair (A, B) , where A is an n by m matrix and B is an n by p matrix.

2 Specification

```
SUBROUTINE F08ZSF (N, M, P, A, LDA, TAUA, B, LDB, TAUB, WORK, LWORK,      &
                  INFO)
INTEGER                N, M, P, LDA, LDB, LWORK, INFO
COMPLEX (KIND=nag_wp) A(LDA,*), TAUA(min(N,M)), B(LDB,*),          &
                  TAUB(min(N,P)), WORK(max(1,LWORK))
```

The routine may be called by its LAPACK name *zggqrf*.

3 Description

F08ZSF (ZGGQRF) forms the generalized QR factorization of an n by m matrix A and an n by p matrix B

$$A = QR, \quad B = QTZ,$$

where Q is an n by n unitary matrix, Z is a p by p unitary matrix and R and T are of the form

$$R = \begin{cases} \begin{matrix} m \\ n-m \end{matrix} \begin{pmatrix} R_{11} \\ \mathbf{0} \end{pmatrix}, & \text{if } n \geq m; \\ \begin{matrix} n & m-n \\ n \end{matrix} \begin{pmatrix} R_{11} & R_{12} \end{pmatrix}, & \text{if } n < m, \end{cases}$$

with R_{11} upper triangular,

$$T = \begin{cases} \begin{matrix} p-n & n \\ n \end{matrix} \begin{pmatrix} \mathbf{0} & T_{12} \end{pmatrix}, & \text{if } n \leq p, \\ \begin{matrix} n-p & p \\ p \end{matrix} \begin{pmatrix} T_{11} \\ T_{21} \end{pmatrix}, & \text{if } n > p, \end{cases}$$

with T_{12} or T_{21} upper triangular.

In particular, if B is square and nonsingular, the generalized QR factorization of A and B implicitly gives the QR factorization of $B^{-1}A$ as

$$B^{-1}A = Z^H(T^{-1}R).$$

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Anderson E, Bai Z and Dongarra J (1992) Generalized *QR* factorization and its applications *Linear Algebra Appl. (Volume 162–164)* 243–271

Hammarling S (1987) The numerical solution of the general Gauss-Markov linear model *Mathematics in Signal Processing* (eds T S Durrani, J B Abbiss, J E Hudson, R N Madan, J G McWhirter and T A Moore) 441–456 Oxford University Press

Paige C C (1990) Some aspects of generalized *QR* factorizations . *In Reliable Numerical Computation* (eds M G Cox and S Hammarling) 73–91 Oxford University Press

5 Arguments

- 1: N – INTEGER *Input*
On entry: n , the number of rows of the matrices A and B .
Constraint: $N \geq 0$.
- 2: M – INTEGER *Input*
On entry: m , the number of columns of the matrix A .
Constraint: $M \geq 0$.
- 3: P – INTEGER *Input*
On entry: p , the number of columns of the matrix B .
Constraint: $P \geq 0$.
- 4: A(LDA,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, M)$.
On entry: the n by m matrix A .
On exit: the elements on and above the diagonal of the array contain the $\min(n, m)$ by m upper trapezoidal matrix R (R is upper triangular if $n \geq m$); the elements below the diagonal, with the array TAUA, represent the unitary matrix Q as a product of $\min(n, m)$ elementary reflectors (see Section 3.3.6 in the F08 Chapter Introduction).
- 5: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08ZSF (ZGGQRF) is called.
Constraint: $LDA \geq \max(1, N)$.
- 6: TAUA(min(N, M)) – COMPLEX (KIND=nag_wp) array *Output*
On exit: the scalar factors of the elementary reflectors which represent the unitary matrix Q .
- 7: B(LDB,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array B must be at least $\max(1, P)$.
On entry: the n by p matrix B .
On exit: if $n \leq p$, the upper triangle of the subarray $B(1 : n, p - n + 1 : p)$ contains the n by n upper triangular matrix T_{12} .

If $n > p$, the elements on and above the $(n - p)$ th subdiagonal contain the n by p upper trapezoidal matrix T ; the remaining elements, with the array TAUB, represent the unitary matrix Z as a product of elementary reflectors (see Section 3.3.6 in the F08 Chapter Introduction).

- 8: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F08ZSF (ZGGQRF) is called.
Constraint: $LDB \geq \max(1, N)$.
- 9: TAUB($\min(N, P)$) – COMPLEX (KIND=nag_wp) array *Output*
On exit: the scalar factors of the elementary reflectors which represent the unitary matrix Z .
- 10: WORK($\max(1, LWORK)$) – COMPLEX (KIND=nag_wp) array *Workspace*
On exit: if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.
- 11: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08ZSF (ZGGQRF) is called.
 If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.
Suggested value: for optimal performance, $LWORK \geq \max(N, M, P) \times \max(nb1, nb2, nb3)$, where $nb1$ is the optimal **block size** for the QR factorization of an n by m matrix, $nb2$ is the optimal **block size** for the RQ factorization of an n by p matrix, and $nb3$ is the optimal **block size** for a call of F08AUF (ZUNMQR).
Constraint: $LWORK \geq \max(1, N, M, P)$ or LWORK = -1.
- 12: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed generalized QR factorization is the exact factorization for nearby matrices $(A + E)$ and $(B + F)$, where

$$\|E\|_2 = O\epsilon\|A\|_2 \quad \text{and} \quad \|F\|_2 = O\epsilon\|B\|_2,$$

and ϵ is the *machine precision*.

8 Parallelism and Performance

F08ZSF (ZGGQRF) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08ZSF (ZGGQRF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The unitary matrices Q and Z may be formed explicitly by calls to F08ATF (ZUNGQR) and F08CWF (ZUNGRQ) respectively. F08AUF (ZUNMQR) may be used to multiply Q by another matrix and F08CXF (ZUNMRQ) may be used to multiply Z by another matrix.

The real analogue of this routine is F08ZEF (DGGQRF).

10 Example

This example solves the general Gauss–Markov linear model problem

$$\min_x \|y\|_2 \quad \text{subject to} \quad d = Ax + By$$

where

$$A = \begin{pmatrix} 0.96 - 0.81i & -0.03 + 0.96i & -0.91 + 2.06i \\ -0.98 + 1.98i & -1.20 + 0.19i & -0.66 + 0.42i \\ 0.62 - 0.46i & 1.01 + 0.02i & 0.63 - 0.17i \\ 1.08 - 0.28i & 0.20 - 0.12i & -0.07 + 1.23i \end{pmatrix},$$

$$B = \begin{pmatrix} 0.5 - 1.0i & 0 & 0 & 0 \\ 0 & 1.0 - 2.0i & 0 & 0 \\ 0 & 0 & 2.0 - 3.0i & 0 \\ 0 & 0 & 0 & 5.0 - 4.0i \end{pmatrix}$$

and

$$d = \begin{pmatrix} 6.00 - 0.40i \\ -5.27 + 0.90i \\ 2.72 - 2.13i \\ -1.30 - 2.80i \end{pmatrix}.$$

The solution is obtained by first computing a generalized QR factorization of the matrix pair (A, B) . The example illustrates the general solution process, although the above data corresponds to a simple weighted least squares problem.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

10.1 Program Text

```
Program f08zsfe
```

```
!      F08ZSF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
!      Use nag_library, Only: dznrm2, nag_wp, zgemv, zggqrf, ztrtrs, zunmqr, &
!                               zunmrq
!      .. Implicit None Statement ..
!      Implicit None
!      .. Parameters ..
!      Complex (Kind=nag_wp), Parameter :: one = (1.0E0_nag_wp,0.0E0_nag_wp)
!      Complex (Kind=nag_wp), Parameter :: zero = (0.0E0_nag_wp,0.0E0_nag_wp)
!      Integer, Parameter :: nb = 64, nin = 5, nout = 6
```

```

! .. Local Scalars ..
Real (Kind=nag_wp)          :: rnorm
Integer                     :: i, info, lda, ldb, lwork, m, n, p
! .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: a(:,,:), b(:,,:), d(:), taua(:),      &
                                         taub(:), work(:), y(:)
! .. Intrinsic Procedures ..
Intrinsic                   :: max, min
! .. Executable Statements ..
Write (nout,*) 'F08ZSF Example Program Results'
Write (nout,*)
! Skip heading in data file
Read (nin,*)
Read (nin,*) n, m, p
lda = n
ldb = n
lwork = nb*(m+p)
Allocate (a(lda,m),b(ldb,p),d(n),taua(m),taub(m+p),work(lwork),y(p))

! Read A, B and D from data file
Read (nin,*)(a(i,1:m),i=1,n)
Read (nin,*)(b(i,1:p),i=1,n)
Read (nin,*) d(1:n)

! Compute the generalized QR factorization of (A,B) as
!  $A = Q(R)$ ,  $B = Q(T11 T12)*Z$ 
!  $(0) \quad (0 T22)$ 
! The NAG name equivalent of zggqrf is f08zsf
Call zggqrf(n,m,p,a,lda,taua,b,ldb,taub,work,lwork,info)

! Compute  $c = (c1) = (Q^*H)*d$ , storing the result in D
!  $(c2)$ 
! The NAG name equivalent of zunmqr is f08auf
Call zunmqr('Left','Conjugate transpose',n,1,m,a,lda,taua,d,n,work,      &
            lwork,info)

! Putting  $Z*y = w = (w1)$ , set  $w1 = 0$ , storing the result in Y1
!  $(w2)$ 
y(1:m+p-n) = zero

If (n>m) Then

! Copy c2 into Y2

y(m+p-n+1:p) = d(m+1:n)

! Solve  $T22*w2 = c2$  for  $w2$ , storing result in Y2

! The NAG name equivalent of ztrtrs is f07tsf
Call ztrtrs('Upper','No transpose','Non-unit',n-m,1,b(m+1,m+p-n+1),      &
            ldb,y(m+p-n+1),n-m,info)

! If (info>0) Then
! Write (nout,*)
! 'The upper triangular factor, T22, of B is singular, '
! Write (nout,*) 'the least squares solution could not be computed'
! Go To 100
! End If

! Compute estimate of the square root of the residual sum of
! squares norm(y) = norm(w2)
! The NAG name equivalent of dznrm2 is f06jjf
rnorm = dznrm2(n-m,y(m+p-n+1),1)

! Form  $c1 - T12*w2$  in D
! The NAG name equivalent of zgemv is f06saf
Call zgemv('No transpose',m,n-m,-one,b(1,m+p-n+1),ldb,y(m+p-n+1),1,      &
            one,d,1)
End If

! Solve  $R*x = c1 - T12*w2$  for x

```

```

!       The NAG name equivalent of ztrtrs is f07tsf
       Call ztrtrs('Upper','No transpose','Non-unit',m,1,a,lda,d,m,info)

       If (info>0) Then
           Write (nout,*) 'The upper triangular factor, R, of A is singular, '
           Write (nout,*) 'the least squares solution could not be computed'
       Else

!       Compute y = (Z**H)*w
!       The NAG name equivalent of zunmrq is f08cxf
       Call zunmrq('Left','Conjugate transpose',p,1,min(n,p),b(max(1,      &
           n-p+1),1),ldb,taub,y,p,work,lwork,info)

!       Print least squares solution x
       Write (nout,*) 'Generalized least squares solution'
       Write (nout,99999) d(1:m)

!       Print residual vector y
       Write (nout,*)
       Write (nout,*) 'Residual vector'
       Write (nout,99998) y(1:p)

!       Print estimate of the square root of the residual sum of
!       squares
       Write (nout,*)
       Write (nout,*) 'Square root of the residual sum of squares'
       Write (nout,99997) rnorm
       End If
100    Continue

99999 Format (3(' (',F9.4,',',F9.4,')',:))
99998 Format (3(' (',1P,E9.2,',',1P,E9.2,')',:))
99997 Format (1X,1P,E10.2)
       End Program f08zsf

```

10.2 Program Data

F08ZSF Example Program Data

```

      4           3           4                               :Values of N, M and P

( 0.96,-0.81) (-0.03, 0.96) (-0.91, 2.06)
(-0.98, 1.98) (-1.20, 0.19) (-0.66, 0.42)
( 0.62,-0.46) ( 1.01, 0.02) ( 0.63,-0.17)
( 1.08,-0.28) ( 0.20,-0.12) (-0.07, 1.23)                               :End of matrix A

( 0.50,-1.00) ( 0.00, 0.00) ( 0.00, 0.00) ( 0.00, 0.00)
( 0.00, 0.00) ( 1.00,-2.00) ( 0.00, 0.00) ( 0.00, 0.00)
( 0.00, 0.00) ( 0.00, 0.00) ( 2.00,-3.00) ( 0.00, 0.00)
( 0.00, 0.00) ( 0.00, 0.00) ( 0.00, 0.00) ( 5.00,-4.00) :End of matrix B

( 6.00,-0.40)
(-5.27, 0.90)
( 2.72,-2.13)
(-1.30,-2.80)                               :End of vector d

```

10.3 Program Results

F08ZSF Example Program Results

```

Generalized least squares solution
( -0.9846,  1.9950) (  3.9929, -4.9748) ( -3.0026,  0.9994)

Residual vector

```

(1.26E-04,-4.66E-04) (1.11E-03,-8.61E-04) (3.84E-03,-1.82E-03)
(2.03E-03, 3.02E-03)

Square root of the residual sum of squares
5.79E-03
